

MATHEMATIK für die ASTRONOMIE

KULTURBUND DER DDR

Zentrale Kommission Astronomie und Raumfahrt – Arbeitskreis „Numerische Astronomie“

Veröffentlichung 7

Leipzig 1988

Liebe Abonnenten dieser Veröffentlichungsreihe!

Der Kreis der Interessenten für die Veröffentlichungen des Arbeitskreises "NUMERISCHE ASTRONOMIE" hat sich bei etwa 250 stabilisiert. Der Arbeitsaufwand beim Versand ist recht erheblich. Besonders ärgerlich ist es, wenn einige Bundesfreunde die Nachnahmesendungen dann nicht von der Post abholen.

Ich bitte alle Freunde der numerischen Astronomie, die weiterhin an unserer Veröffentlichung interessiert sind, zehn Mark im Voraus an mich zu schicken. Wer mehr als ein Exemplar wünscht, muß entsprechend mehr Vorkasse leisten.

Diese Form des Versands ist billiger und bequemer für alle wirklichen Interessenten. Wenn die zehn Mark aufgebraucht sind, erscheint ein Hinweis in diesem Heft. Der Preis pro Exemplar wird weiterhin bei etwa 1.50 M bis 2.50 M liegen. Es fallen jedoch die Nachnahmegebühren und die Gebühren für die Rücksendung der Zahlkarte weg.

Versandadresse wie bisher: Günter Bethge Kreuzstraße 3B LEIPZIG 7010

Das Geld sollte innerhalb von zwei Monaten eingegangen sein.

Es ist eine traditionelle Aufgabe der Astronomie, den Kalender zu erstellen. Diese Veröffentlichung bietet ein leicht verständliches und gut nachvollziehbares BASIC-Programm, das auf allen grafikfähigen Rechnern läuft.

Das Programm bietet außer den Feiertagen auch die Möglichkeit, ganz persönliche sogenannte Merktage ausgeben zu lassen. Es ergibt sich daraus die Möglichkeit einer ganz alltäglichen Nutzung.

Kalender mit persönlichen Daten

Ab der Zeile 1000 kann das Programm ohne Änderungen übernommen werden. Alle rechnerartigen Operationen liegen in den Zeilen darunter.

Anhand der vier Beispiele: (SPECTRUM, C64, C128 u. MS-DOS-Rechner) ist es leicht, das Programm an jeden anderen Computer anzupassen.

Die meisten Rechner lassen Zuweisungen ohne das Schlüsselwort LET zu. Ich habe alle Zuweisungen unterstrichen dargestellt. Wenn nötig, ist dort nur ein LET voranzustellen (z.B. SPECTRUM).

Wenn Druckertreiber zu laden sind, so muß das jeder selbst ins Programm einfügen.

Am Ende des Programms stehen in DATA- Zeilen die Feier- und Merktage. Bei Änderungen muß darauf geachtet werden, daß jeweils am Ende der drei Rubriken eine Null steht.

Auch ohne Drucker ist dieses Programm durchaus nützlich.

Einige Erklärungen zum Programmaufbau

Ab der Zeile 1170 werden die DATA-Zeilen durchsucht und die Daten gezählt, die für das aktuelle Jahr gebraucht werden. In Zeile 190 werden dann alle Felder dimensioniert. Vorgesehen sind:

- M(32,a) - Monatsfeld zum aktuellen Monat
- a=1 - Julianisches Datum des Tages
- a=2 - Wochentag des Tages (0 bis 6)
- a=3 - wenn Inhalt =0 - kein Merks-tag; wenn =1 - Merks-tag
- a=4 - wenn Inhalt =0 - kein Feiertag; wenn =1 - Feiertag

Auf Index 0 wurde verzichtet, weil auf manchen Rechnern nicht existent.

- T (I+7) - I=Merks-tage aus DATA; +7 variable Merks-tage
- D\$(I+7) - dazugehöriger Text
- G (J+2) / F\$(J+2) - wie oben - nur Feiertage

- C\$ (b) - Text, der am unteren Ende des Kalenders ausgegeben wird
- L (b) / S (b) - Zeile und Spalte der Ausgabe von C\$ (b)

- O\$ (12) - "JANUAR" bis "DEZEMBER"

Ab den Zeilen 3000 bzw. 3300 werden die Felder G(I) bzw. T(I) bestückt und anschließend sortiert.

Ab der Zeile 1800 wird getestet, wieviel Tage der Monat hat. Dabei werden in das Monatsfeld die Julian. Daten und die Wochentage eingetragen.

Ab 2000 wird überprüft, ob Merk- oder/und Feiertage vorliegen und es werden entsprechende Flags gesetzt. In 2110 werden die Arbeitstage gezählt.

Obwohl die Zeiger auf den Beginn der Durchsuchung ständig aktualisiert werden, dauert das Durchsuchen, je nach Rechnertyp relativ lange. Es wird deshalb zuerst das leere Kalenderblatt gezeichnet.

Nach Fertigstellung des aktuellen Monats kommt man mit einem beliebigen Tastendruck in ein kleines Menue, das durchaus erweitert werden kann. (Abspeichern des Blattes auf Datenträger, Ausdrucken verschiedener Formate u.s.w.)

Bei Erreichen des Dezembers (Zeile 1410) schließt das Programm ab. Es ist dort auch ein Neustart möglich (statt GOTO 740 dann GOTO 730). Das ist aber in kompilierten Programmen nicht möglich, da dort nur eine einmalige Dimensionierung erlaubt ist.

Statt DATA-Zeilen ist auch ein Ablegen der Daten auf Datenträger möglich. Das hat den Vorteil, daß man für verschiedene Zwecke verschiedene Kalender mit dem gleichen Programm erstellen kann.

Anmerkung: Die Zeile 1610 ist nur im SPECTRUM-Emulator des C64 nötig, da dort $-1 * 0 = -1E-38$ und nicht 0 ergibt.

Zunächst folgt das Hauptprogramm und die Unterprogramme, die auf allen Rechnern ohne Änderungen laufen.

```
1000 REM
1010 REM *****
1020 REM *
1030 REM *   HAUPTPROGRAMM   *
1040 REM *
1050 REM *****
1060 REM
1070 GOSUB 250 : GOSUB 230 : GOSUB 220
1080 PRINT : PRINT "KALENDER MIT PERSOENLICHEN DATEN" : PRINT
1090 PRINT "NUR IM GREGORIANISCHEN KALENDER" : PRINT
1100 PRINT "----- VERSION 3.2 -----" : PRINT
1110 PRINT "PROGRAMMIERER:   GUENTER BETHGE" : PRINT
1120 PRINT "KREUZSTRASSE 3B   LEIPZIG 7010" : PRINT
1130 INPUT "JAHR           ";JR : PRINT
1140 IF JR<1583 THEN PRINT "ERST AB 1583 !!" : GOSUB 170 : GOTO 730
1150 INPUT "STARTMONAT ";M0 : IF M0<1 OR M0>12 THEN : GOTO 1150
1160 PRINT : PRINT "WARTEN ..." : I=0 : J=0
1170 READ Z : IF Z=0 THEN GOTO 1190
1180 READ Z : READ Z$ : J=J+1 : GOTO 1170
1190 READ Z : IF Z=0 THEN GOTO 1210
1200 READ Z : READ Z$ : I=I+1 : GOTO 1190
1210 READ Z : IF Z=0 THEN GOTO 1240
1220 IF Z<>JR THEN READ Z : READ Z : READ Z$ : GOTO 1210
1230 READ Z : READ Z : READ Z$ : I=I+1 : GOTO 1210
1240 GOSUB 190 : RESTORE
1250 M=1 : T=1 : F=1 : REM GREGORIANISCHER KALENDER
1260 GOSUB 2200 : GOSUB 3000 : GOSUB 3300 : GOSUB 250 : MT=M-1
1270 MT=MT+1 : GOSUB 260 : GOSUB 280 : GOSUB 400 : GOSUB 1800 : GOSUB 2000
1280 GOSUB 290 : GOSUB 610
1290 GOSUB 170 : GOSUB 230 : GOSUB 220
1300 PRINT : PRINT : PRINT
1310 PRINT "AUSDRUCKEN           1"
1320 PRINT
1330 PRINT "ENDE                   2"
1340 PRINT : PRINT
1350 PRINT "MIT   LEERTASTE   WEITER"
1360 GOSUB 170
1370 IF A$="1" THEN GOSUB 270 : GOTO 1290
1380 IF A$="2" THEN GOTO 740
1390 IF A$=" " THEN GOSUB 720 : GOTO 1410
1400 GOTO 1360
1410 IF MT=12 THEN GOTO 740
1420 GOSUB 260 : GOTO 1270
1430 REM
1440 REM *****
1450 REM *
1460 REM *   HAUPTPROGRAMMENDE *
1470 REM *
1480 REM *****
1490 REM
```

```

1500 REM TG=TAG MT=MONAT JR=JAHR
1510 REM
1520 REM JULIANISCHES DATUM ****
1530 REM
1540 REM JD=JUL.DAT. WG=WOCHENT
1550 REM
1600 JD=-INT(7*(INT((MT+9)/12)+JR)/4) : S0=SGN(MT-9) : A=INT(ABS(MT-9)/7)
1610 IF A=0 THEN S0=0 : REM NUR IM SPECTRUM-EMULATOR DES C64 NOETIG
1620 J1=INT(JR+S0*A) : J1=-INT((INT(J1/100)+1)*3/4)
1630 JD=JD+INT(275*MT/9)+TG+F1*J1 : JD=JD+1721026.5+2*F1+367*JR
1640 A=JD+1.5 : B=A/7 : WG=A-INT(B)*7
1650 RETURN
1660 M(TT,1)=JD : M(TT,2)=WG : RETURN
1670 REM
1680 REM
1690 REM
1700 REM --> JR=JAHR -- MT=MONAT
1710 REM
1720 REM TAGE ZUM AKT.MONAT ****
1730 REM
1740 REM <--- MO(T,W)=TAGEFELD
1750 REM
1800 MS=MT : JS=JR : TG=1 : TT=1 : GOSUB 1490 : GOSUB 1660
1810 TT=TT+1 : JD=JD+1 : GOSUB 2450 : GOSUB 1660 : IF TT<28 THEN GOTO 1810
1820 TT=TT+1 : JD=JD+1 : GOSUB 2400 : GOSUB 1660 : IF MT<>MS THEN GOTO 1840
1830 GOTO 1820
1840 M0=TT-1 : MT=MS : JR=JS : RETURN
1850 REM
1860 REM
1870 REM
1880 REM -> JR=JAHR -- MT=MONAT
1890 REM
1900 REM TAGEVERGLEICH *****
1910 REM
1920 REM <- M(I,4)=1 -> FT.OD.SO
1930 REM <- M(I,3)=1 -> MERKTAG
1940 REM <- AE = ARBEITSTAGE
1950 REM
2000 K=0
2010 AE=0 : FOR I=1 TO M0 : FOR J=G1 TO G0
2020 IF M(I,1)<G(J) THEN M(I,4)=0 : GOTO 2060
2030 IF M(I,1)>G(J) THEN GOTO 2050
2040 M(I,4)=1 : TG=I : AS=FS(J) : K=K+1 : GOSUB 600 : G1=J : GOTO 2060
2050 NEXT J
2060 FOR J=T1 TO T0
2070 IF M(I,1)<T(J) THEN M(I,3)=0 : GOTO 2110
2080 IF M(I,1)>T(J) THEN GOTO 2100
2090 M(I,3)=1 : TG=I : AS=DS(J) : K=K+1 : GOSUB 600 : T1=J : GOTO 2110
2100 NEXT J
2110 CH=K : IF M(I,4)=0 AND M(I,2)<>6 AND M(I,2)<>0 THEN AE=AE+1
2120 NEXT I : RETURN

```

```

2130 REM
2140 REM
2150 REM
2160 REM MONATSSTRINGS *****
2170 REM
2200 Q$(1)="JANUAR" : Q$(2)="FEBRUAR" : Q$(3)="MÄRZ" : Q$(4)="APRIL"
2210 Q$(5)="MAI" : Q$(6)="JUNI" : Q$(7)="JULI" : Q$(8)="AUGUST"
2220 Q$(9)="SEPTEMBER" : Q$(10)="OKTOBER" : Q$(11)="NOVEMBER"
2230 Q$(12)="DEZEMBER" : RETURN
2240 REM
2250 REM
2260 REM
2270 REM > JD=JULIANISCHES DATUM
2280 REM
2290 REM BÜRGERLICHES DATUM ***
2300 REM
2310 REM TG=Tag MT=Monat JR=Jahr
2320 REM
2400 J1=JD+5 : IF F1=0 THEN A=J1 : GOTO 2420
2410 A1=INT((J1/36524.25)-51.12264) : A=J1+1+A1-INT(A1/4)
2420 B=A+1524;C0=INT((B/365.25)-.3343);T5=INT(365.25*C0);E=INT((B-T5)/30.61)
2430 TG=B-T5-INT(30.61*E) : MT=E-1 : JR=C0-4716 : IF E>13.5 THEN MT=MT-12
2440 IF MT<2.5 THEN JR=JR+1
2450 A=JD+1.5 : B=A/7 : WG=A-INT(B)*7 : RETURN
2460 REM
2470 REM
2480 REM
2490 REM ---> JR=JAHR
2500 REM
2510 REM OSTERN *****
2520 REM
2530 REM <- OM=MONAT OT=SONNTAG
2540 REM
2600 IF JR<1583 THEN STOP
2610 GOTO 2630
2620 QU=DI/DV : QG=INT(QU) : RE=DI-QG*DV : RETURN
2630 DI=JR : DV=19 : GOSUB 2620 : AA=RE
2640 DV=100 : GOSUB 2620 : BB=QG : CC=RE
2650 DI=BB : DV=4 : GOSUB 2620 : DD=QG : EE=RE
2660 DI=BB+8 : DV=25 : GOSUB 2620 : FF=QG
2670 DI=BB-FF+1 : DV=3 : GOSUB 2620 : GG=QG
2680 DI=19*AA+BB-DD-GG+15 : DV=30 : GOSUB 2620 : HH=RE
2690 DI=CC : DV=4 : GOSUB 2620 : II=QG : KK=RE
2700 DI=32+2*EE+2*II-HH-KK : DV=7 : GOSUB 2620 : LL=RE
2710 DI=AA+11*HH+22*LL : DV=451 : GOSUB 2620 : MM=QG
2720 DI=HH+LL-7*MM+114 : DV=31 : GOSUB 2620 : OM=QG : OT=RE+1
2730 RETURN

```

Das Ergebnis aus der Berechnung des Ostersonntags bildet die Basis zur Berechnung aller, von Ostern abhängiger Feier- und Merktage.

```

2740 REM
2750 REM
2760 REM
2770 REM ----> JR=JAHR -- G0
2780 REM
2790 REM FEIERTAGE IM AKT. JAHR *
2800 REM
2810 REM <-- G(I)=JD F$(I)=TEXT
2820 REM
3000 I=0
3010 I=I+1 : READ TG : IF TG=0 THEN GOTO 3030
3020 READ MT : GOSUB 1600 : G(I)=JD : READ F$(I) : GOTO 3010
3030 GOSUB 2600 : TG=OT : MT=OM : GOSUB 1600 : G(I)=JD-2
3040 F$(I)= "KARFREITAG " : I=I+1 : G(I)=JD+50
3050 F$(I)= "PFINGSTMONTAG " : G0=I
3060 FOR J=G0 TO 1 STEP -1 : F0=0 : FOR I=2 TO J
3070 IF G(I-1)<=G(I) THEN GOTO 3100
3080 A=G(I-1) : G(I-1)=G(I) : G(I)=A : F0=1
3090 A$=F$(I-1) : F$(I-1)=F$(I) : F$(I)=A$
3100 NEXT I : IF F0=0 THEN GOTO 3120
3110 NEXT J
3120 RETURN
3130 REM
3140 REM
3150 REM
3160 REM ---> JR=JAHR --- T0
3170 REM
3180 REM MERKTAGE IM AKT. JAHR **
3190 REM
3200 REM <-- T(I)=JD D$(I)=TEXT
3210 REM
3300 I=0
3310 I=I+1 : READ TG : IF TG=0 THEN GOTO 3330
3320 READ MT : GOSUB 1600 : T(I)=JD : READ D$(I) : GOTO 3310
3330 GOSUB 2600 : TG=OT : MT=OM : GOSUB 1600 : T(I)=JD+39
3340 D$(I)= "HIMMELFAHRT " : I=I+1 : T(I)=JD-48
3350 D$(I)= "ROSENMONTAG " : I=I+1 : T(I)=JD-47
3360 D$(I)= "FASTNACHT " : I=I+1 : T(I)=JD-46
3370 D$(I)= "ASCHERMITTW. " : I=I+1
3380 TG=24 : MT=12 : GOSUB 1600 : JD=JD-WG : T(I)=JD-21
3390 D$(I)= "1. ADVENT " : I=I+1 : T(I)=JD-28
3400 D$(I)= "TOTENSONNTAG " : I=I+1 : T(I)=JD-32
3410 D$(I)= "BUSSTAG "
3420 REM
3430 REM JAHRESGEBUNDENE MERKT.
3440 JJ=JR
3450 I=I+1 : READ JR : IF JR=0 THEN GOTO 3480
3460 IF JR<>JJ THEN I=I-1 : READ A : READ A : READ A$ : GOTO 3450
3470 READ TG : READ MT : GOSUB 1600 : T(I)=JD : READ D$(I) : GOTO 3450
3480 JR=JJ : T0=I-1
3490 FOR J=T0 TO 1 STEP -1 : F0=0 : FOR I=2 TO J
3500 IF T(I-1)<=T(I) THEN GOTO 3530
3510 A=T(I-1) : T(I-1)=T(I) : T(I)=A : F0=1
3520 A$=D$(I-1) : D$(I-1)=D$(I) : D$(I)=A$
3530 NEXT I : IF F0=0 THEN GOTO 3550
3540 NEXT J
3550 RETURN

```

```

4000 REM
4010 REM
4020 REM
4030 REM FEIERTAGE *****
4040 REM
4050 DATA 1, 1, "NEUJAHR      "
4060 DATA 1, 5, "KAMPFT. D. ARB."
4070 DATA 7, 10, "TAG D. REPUBL."
4080 DATA 25, 12, "1. FEIERTAG  "
4090 DATA 26, 12, "2. FEIERTAG  "
4990 DATA 0
5000 REM
5010 REM
5020 REM
5030 REM MERKTAGE F. JEDES JAHR *
5040 REM
5050 DATA 12, 6, "LEHRERTAG      "
5060 DATA 6, 12, "ULRIKES GEBT."
5070 DATA 7, 3, "HOCHZEITSTAG "
5080 DATA 24, 12, "HEILIGABEND  "
5090 DATA 31, 12, "SILVESTER    "
5100 DATA 1, 6, "KINDERTAG     "
5110 DATA 1, 9, "WELTFRIEDENST"
5990 DATA 0
6000 REM
6010 REM
6020 REM
6030 REM MERKT. M. JAHRESANGABE **
6040 REM
6050 DATA 1988, 27, 3, "MESZ BEGINN  "
6060 DATA 1988, 25, 9, "MESZ ENDE    "
6070 DATA 1988, 13, 3, "MESSEBEGINN "
6080 DATA 1988, 4, 9, "MESSEBEGINN "
6090 DATA 1989, 26, 3, "MESZ BEGINN "
6100 DATA 1989, 24, 9, "MESZ ENDE    "
6110 DATA 1988, 8, 10, "SCHULFREI    "
6120 DATA 1988, 13, 10, "FERIENBEGINN "
6130 DATA 1988, 24, 10, "SCHULBEGINN "
6140 DATA 1988, 23, 12, "FERIENBEGINN "
6150 DATA 1989, 3, 1, "SCHULBEGINN "
6160 DATA 1989, 4, 2, "FERIENBEGINN "
6170 DATA 1989, 27, 2, "SCHULBEGINN "
6180 DATA 1989, 25, 3, "SCHULFREI    "
6190 DATA 1989, 13, 5, "FERIENBEGINN "
6200 DATA 1989, 22, 5, "SCHULBEGINN "
6210 DATA 1989, 1, 7, "FERIENBEGINN "
6220 DATA 1989, 1, 9, "SCHULBEGINN "
6990 DATA 0

```

Es folgen die Zeilen 100 bis 740 für vier verschiedene Rechner! Bei Anpassung an andere Rechnertypen ist darauf zu achten, daß bestimmte Operationen vom Hauptprogramm in entsprechenden Zeilen erwartet werden.

Im SPECTRUM sind 14 Textausgaben im unteren Teil des Kalenders möglich.

```

100 GOTO 1070
110 REM *****
120 REM *
130 REM * SPECTRUM
140 REM *
150 REM *****
160 REM
170 LET A$=INKEY$ : IF A$="" THEN GOTO 170 : REM WARTEN AUF TASTENDRUCK
180 RETURN
190 DIM M(32,4) : DIM T(I+10) : DIM G(J+2) : DIM D$(I+10,12)
200 DIM F$(J+2,12) : DIM O$(12,9) : DIM C$(15,15) : DIM L(15) : DIM S(15)
210 RETURN
220 CLS : RETURN : REM TEXTSCHIRM LOESCHEN
230 CLS : RETURN : REM TEXTSCHIRM EINSCHALTEN
240 REM GRAFIKSCHIRM EINSCHALTEN, LOESCHEN UND FARBEN SETZEN
250 CLS : RETURN
260 CLS : RETURN : REM GRAFIKSCH. EINSCH. U. LOESCHEN
265 REM HARDCOPY EINES BLATTES
270 CLS : GOSUB 280 : GOSUB 400 : GOSUB 290 : GOSUB 610 : COPY : RETURN
280 PRINT AT 0,0;O$(MT) : PRINT AT 0,28;STR$(JR) : RETURN
290 PRINT AT 0,11;STR$(AE) : RETURN
370 REM
380 REM LEERES KALENDERBLATT ZEICHNEN *
390 REM
400 PLOT 0,159 : DRAW 0,-159 : DRAW 255,0 : DRAW 0,159
410 PRINT AT 0,14;"ARBEITSTAGE" : INVERSE 1
420 PRINT AT 1,0;"SO MO DI MI DO FR SA " : INVERSE 0
430 PLOT 0,143 : DRAW 255,0 : PLOT 0,127 : DRAW 255,0 : PLOT 0,111 : DRAW 255,0
440 PLOT 0,95 : DRAW 255,0 : PLOT 0,79 : DRAW 255,0 : PLOT 0,63 : DRAW 63,0
450 PLOT 23,159 : DRAW 0,-96 : PLOT 63,159 : DRAW 0,-96 : PLOT 103,159 : DRAW 0,-80
460 PLOT 143,159 : DRAW 0,-80 : PLOT 183,159 : DRAW 0,-80 : PLOT 223,159 : DRAW 0,-80
470 RETURN
570 REM
580 REM KALENDERBLATT AUSFUELLEN *
590 REM
600 LET C$(K)=STR$(TG)+"/"+A$ : IF K=15 THEN LET K=K-1
605 RETURN
610 LET ZE=3 : FOR I=1 TO M0 : LET SP=M(I,2)*5-2 : IF M(I,2)=0 THEN LET SP=SP+2
620 LET PO=0 : IF M(I,2)=0 OR M(I,4)=1 THEN LET PO=1
630 IF PO=0 THEN PRINT AT ZE,SP;I
635 IF PO=1 THEN INVERSE 1:PRINT AT ZE-1,SP;" " : INVERSE 0:PRINT AT ZE,SP;I
640 IF M(I,3)=0 THEN GOTO 660
650 PRINT AT ZE-1,SP+2;"X"
660 IF M(I,2)=6 THEN LET ZE=ZE+2
670 NEXT I
680 FOR I=1 TO 8 : LET L(I)=I+12 : LET S(I)=17 : NEXT I
690 FOR I=9 TO 14 : LET L(I)=I+6 : LET S(I)=0 : NEXT I
700 FOR I=1 TO CH : PRINT AT L(I),S(I);C$(I) : NEXT I
710 GOSUB 400 : RETURN
720 FOR I=1 TO 15 : LET C$(I)="" : NEXT I : RETURN
730 RUN : REM NEUSTART
740 GOSUB 220 : PRINT "PROGRAMMENDE" : STOP : REM ENDE

```


Im C64 mit SIMON'S BASIC sind 11 Textausgaben im unteren Teil des Kalenders möglich.

```
100 GOTO 1070
110 REM *****
120 REM *
130 REM * COMMODORE C64 SIMON'S BASIC *
140 REM *
150 REM *****
160 REM
170 GET A$: IF A$="" THEN GOTO 170 : REM WARTEN AUF TASTENDRUCK
180 RETURN
190 REM VARIABLE DIMENSIONIERUNG
200 DIM M(32,4),T(I+7),G(J+2),D$(I+7),F$(J+2),O$(12),C$(12),L(12),S(12)
210 RETURN
220 PRINT CHR$(147) : RETURN : REM TEXTSCHIRM LOESCHEN
230 NRM : RETURN : REM TEXTSCHIRM EINSCHALTEN
240 REM GRAFIKSCHIRM EINSCHALTEN, LOESCHEN UND FARBEN SETZEN
250 HIRES 2,7 : POKE 53280,7 : POKE 53281,7 : POKE 646,0 : RETURN
260 HIRES 2,7 : RETURN : REM GRAFIKSCH. EINSCH. U. LOESCHEN
270 CSET 2 : COPY : RETURN : REM AUSDRUCKEN EINES KAL-BLATTES
280 TEXT 0,0,(O$(MT)),1,1,8 : TEXT 280,0,(STR$(JR)),1,1,8 : RETURN
290 TEXT 108,0,(STR$(AE)),1,1,8 : RETURN
370 REM
380 REM LEERES KALENDERBLATT ZEICHNEN *
390 REM
400 TEXT 140,0,"ARBEITSTAGE",1,1,8
410 TEXT 1,9,"SO MO DI MI DO FR SA ",1,1,8
420 LINE 0,7,319,7,1 : LINE 319,7,319,199,1 : LINE 319,199,0,199,1
430 LINE 0,199,0,7,1 : LINE 0,16,319,16,1
440 LINE 0,40,319,40,1 : LINE 0,64,319,64,1 : LINE 0,88,319,88,1
450 LINE 0,112,319,112,1 : LINE 0,136,319,136,1 : LINE 0,160,86,160,1
460 LINE 38,7,38,160,1 : LINE 86,7,86,160,1 : LINE 134,7,134,136,1
470 LINE 182,7,182,136,1 : LINE 230,7,230,136,1 : LINE 278,7,278,136,1
480 RETURN
570 REM
580 REM KALENDERBLATT AUSFUELLEN *
590 REM
600 C$(K)=RIGHT$(STR$(TG),2)+"/ "+A$ : IF K=12 THEN K=K-1
605 RETURN
610 ZE=32 : FOR I=1 TO M0 : SP=M(I,2)*48-8 : IF M(I,2)=0 THEN SP=SP+8
620 IF M(I,4)=0 AND M(I,2)<>0 THEN GOTO 640
630 TEXT SP,ZE,(CHR$(18)+" "+CHR$(146)),1,1,8
640 TEXT SP,ZE,RIGHT$(STR$(I),2),2,1,8
650 IF M(I,3)=0 THEN GOTO 670
660 TEXT SP+24,ZE-8,"X",1,1,8
670 IF M(I,2)=6 THEN ZE=ZE+24
680 NEXT I
690 FOR I=1 TO 7 : L(I)=132+I*8 : S(I)=168 : NEXT I
700 FOR I=8 TO 11: L(I)=100+I*8 : S(I)=8 : NEXT I
710 FOR I=1 TO CH : TEXT S(I),L(I),(C$(I)),1,1,8 : NEXT I:GOSUB 400:RETURN
720 FOR I=0 TO 12 : C$(I)="" : NEXT I : RETURN
730 RUN : REM NEUSTART
740 GOSUB 220 : PRINT "PROGRAMMENDE" : END : REM PROGRAMMENDE
```

Im C128 ist es zweckmäßig, sinnvoll mit FAST und SLOW zu arbeiten. Die COLOR-Anweisungen in Zeile 250 erzeugen im PLUS4 UND C16 einen leeren Bildschirm. Sie müssen geändert oder können einfach weggelassen werden.

Diese Version bietet die Möglichkeit für 9 Textausgaben im unteren Teil des Kalenderblattes.

```

100 GOTO 1070 : REM ZUVOR DRUCKERTREIBER LADEN
110 REM *****
120 REM * *
130 REM * COMMODORE C128/PLUS4/C16 *
140 REM * *
150 REM *****
160 REM
170 GET AS : IF AS="" THEN GOTO 170 : REM WARTEN AUF TASTENDRUCK
180 RETURN
190 REM VARIABLE DIMENSIONIERUNG
200 DIM M(32,4),T(I+7),G(J+2),Ds(I+7),Fs(J+2),Os(12),Cs(10),L(10),S(10)
210 RETURN
220 SCNCLR :RETURN : REM TEXTSCHIRM LOESCHEN
230 GRAPHIC 0 :RETURN : REM TEXTSCHIRM EINSCHALTEN
240 REM FARBEN SETZEN !! (COLOR-ANWEISUNGEN VERSCHIEDEN) !! BEISPIEL C128
250 GRAPHIC 1,1 :COLOR 1,2:COLOR 0,10:COLOR 4,10:COLOR 5,8: RETURN
260 GRAPHIC 1,1 :RETURN : REM GRAFIKSCH. EINSCHALTEN UND LOESCHEN
270 GRAPHIC 1 :RETURN : REM MIT MASCHINENPROGRAMMAUFRUF AUSDRUCKEN
280 CHAR ,0,0,Os(MT) : CHAR ,35,0,STR$(JR) : RETURN
290 CHAR ,13,0,STR$(AE) : RETURN
370 REM
380 REM LEERES KALENDERBLATT ZEICHNEN *
390 REM
400 DRAW , 0,8 TO 319,8 TO 319,199 TO 0,199 TO 0,8
410 CHAR ,17,0,"ARBEITSTAGE"
420 CHAR ,0,1,"SO MO DI MI DO FR SA ",1
430 DRAW,0,39 TO 319,39 : DRAW,0,63 TO 319,63 : DRAW,0,87 TO 319,87
440 DRAW,0,111 TO 319,111: DRAW,0,135 TO 319,135: DRAW,0,159 TO 88,159
450 DRAW,40,16 TO 40,159 : DRAW,88,16 TO 88,159 : DRAW,136,16 TO 136,135
460 DRAW,184,16 TO 184,135: DRAW,232,16 TO 232,135: DRAW,280,16 TO 280,135
470 RETURN
570 REM
580 REM KALENDERBLATT AUSFUELLEN *
590 REM
600 Cs(K)=RIGHT$(STR$(TG),2)+"./ "+AS : IF K=10 THEN K=K-1
605 RETURN
610 ZE=4 : FOR I=1 TO M0 : SP=M(I,2)*6-1 : IF M(I,2)=0 THEN SP=SP+1
620 PO=0 : IF M(I,2)=0 OR M(I,4)=1 THEN PO=1
630 CHAR ,SP,ZE,RIGHT$(STR$(I),2),PO
640 IF M(I,3)=0 THEN GOTO 660
650 CHAR ,SP+3,ZE-1,"X"
660 IF M(I,2)=6 THEN ZE=ZE+3
670 NEXT I
680 FOR I=1 TO 6 : L(I)=I+17 : S(I)=20 : NEXT I
690 FOR I=7 TO 9 : L(I)=I+14 : S(I)=1 : NEXT I
700 FOR I=1 TO CH : CHAR ,S(I),L(I),Cs(I) : NEXT I
710 GOSUB 400 : RETURN
720 FOR I=1 TO 10 : Cs(I)="" : NEXT I : RETURN
730 RUN : REM NEUSTART
740 GOSUB 220 : PRINT "PROGRAMMENDE" : END : REM PROGRAMMENDE

```

Unter MS-DOS im GWBASIC oder TURBO-BASIC (auch als EXE-File) sind 9 Textausgaben im unteren Teil des Kalenders möglich.

```
100 GOTO 1070 : REM VOR PROGRAMMSTART GRAPHICS LADEN
110 REM *****
120 REM *
130 REM * MS-DOS GWBASIC / TURBOBASIC *
140 REM *
150 REM *****
160 REM
170 AS=INKEY$ : IF AS="" THEN GOTO 170 : REM WARTEN AUF TASTENDRUCK
180 RETURN
190 REM VARIABLE DIMENSIONIERUNG
200 DIM M(32,4),T(I+7),G(J+2),D$(I+7),F$(J+2),O$(12),C$(10),L(10),S(10)
210 RETURN
220 SCREEN 1 : CLS : RETURN : REM TEXTSCHIRM LOESCHEN
230 CLS : RETURN : REM TEXTSCHIRM EINSCHALTEN
240 REM GRAFIKSCHIRM EINSCHALTEN, LOESCHEN UND FARBEN SETZEN
250 CLS : RETURN
260 CLS : RETURN : REM GRAFIKSCH. EINSCHALTEN UND LOESCHEN
265 REM HARDCOPY DES KALENDERBLATTES DURCH SHIFT PrtSc
270 CLS : GOSUB 280 : GOSUB 400 : GOSUB 290 : GOSUB 610 : RETURN
280 PRINT O$(MT); : LOCATE 1,36,0 : PRINT STR$(JR); : RETURN
290 LOCATE 1,14,0 : PRINT STR$(AE); : RETURN
370 REM
380 REM LEERES KALENDERBLATT ZEICHNEN *
390 REM
400 LOCATE 1,18,0 : PRINT "ARBEITSTAGE";
410 LOCATE 2,1,0 : PRINT "SO MO DI MI DO FR SA "
420 LINE (0,7)-(319,7) : LINE -(319,199) : LINE -(0,199) : LINE -(0,7)
430 LINE (0,16)-(319,16)
440 LINE (0,40)-(319,40) : LINE (0,64)-(319,64) : LINE (0,88)-(319,88)
450 LINE (0,112)-(319,112):LINE (0,136)-(319,136):LINE (0,160)-(86,160)
460 LINE (38,7)-(38,160) : LINE (86,7)-(86,160) : LINE (134,7)-(134,136)
470 LINE (182,7)-(182,136):LINE (230,7)-(230,136):LINE (278,7)-(278,136)
480 RETURN
570 REM
580 REM KALENDERBLATT AUSFUELLEN *
590 REM
600 C$(K)=RIGHT$(STR$(TG),2)+"/ "+AS: IF K=10 THEN K=K-1
605 RETURN
610 ZE=5 : FOR I=1 TO M0 : SP=M(I,2)*6 : IF M(I,2)=0 THEN SP=SP+1
620 PO=0 : IF M(I,2)=0 OR M(I,4)=1 THEN PO=1
625 IF PO=1 THEN PAINT (SP*8,ZE*8-8),2,3
630 LOCATE ZE,SP,0 : PRINT RIGHT$(STR$(I),2);
640 IF M(I,3)=0 THEN GOTO 660
650 LOCATE ZE-1,SP+3 : PRINT "X"
660 IF M(I,2)=6 THEN ZE=ZE+3
670 NEXT I
680 FOR I=1 TO 6 : L(I)=I+18 : S(I)=21 : NEXT I
690 FOR I=7 TO 9 : L(I)=I+15 : S(I)=2 : NEXT I
700 FOR I=1 TO CH : LOCATE L(I),S(I) : PRINT C$(I); : NEXT I
710 GOSUB 400 : RETURN
720 FOR I=1 TO 10 : C$(I)="" : NEXT I : RETURN
730 RUN : REM NEUSTART
740 GOSUB 220 : PRINT "PROGRAMMENDE" : GOSUB 170
750 SCREEN 2 : SCREEN 0 : END : REM PROGRAMMENDE
```

Zum Abschluß 2 Beispielausdrucke (GWBASIC und C128 bzw. PLUS4 bzw. C16)

MAI 21 ARBEITSTAGE 1989						
SO	MO	DI	MI	DO	FR	SA
	1	2	3	4 X	5	6
7	8	9	10	11	12	13 X
14	15	16	17	18	19	20
21	22 X	23	24	25	26	27
28	29	30	31			
1. / KAMPF D. ARB. 4. / HIMMELFAHRT 13. / FERIENBEGINN 15. / PFINGSTMONTAG 22. / SCHULBEGINN						

MÄRZ 22 ARBEITSTAGE 1989						
SO	MO	DI	MI	DO	FR	SA
			1	2	3	4
	6	7 X	8	9	10	11
	13	14	15	16	17	18
	20	21	22	23	24	25 X
X	27	28	29	30	31	
7. / HOCHZEITSTAG 24. / KARFREITAG 25. / SCHULFREI 26. / MESZ BEGINN						